

# Principles of Recursion and Induction for Nominal Lambda Calculus.

Ana Bove <sup>1</sup>

*Chalmers University of Technology  
Gothenburg, Sweden*

Maribel Fernandez <sup>2</sup>

*King's College London  
London, England*

Álvaro Tasistro <sup>3</sup> Nora Szasz <sup>4</sup> Ernesto Copello <sup>5</sup>

*Universidad ORT Uruguay  
Montevideo, Uruguay*

---

## Abstract

We formulate principles of induction and recursion for a variant of lambda calculus with bound names where  $\alpha$ -conversion is based upon name swapping as in nominal abstract syntax. The principles allow to work modulo alpha-conversion and apply the Barendregt variable convention. We derive them all from the simple structural induction principle and apply them to get some fundamental meta-theoretical results, such as the substitution lemma for alpha-conversion and the result of substitution composition. The whole work is implemented in Agda.

*Keywords:* Formal Metatheory, Lambda Calculus, Constructive Type Theory

---

All the shown code is compiled in the last Agda's version 2.4.2.2 and 0.9 standard library, and can be fully accessed at:

<https://github.com/ernius/formalmetatheory-nominal>

---

<sup>1</sup> Email: [bove@chalmers.se](mailto:bove@chalmers.se)

<sup>2</sup> Email: [Maribel.Fernandez@kcl.ac.uk](mailto:Maribel.Fernandez@kcl.ac.uk)

<sup>3</sup> Email: [tasistro@ort.edu.uy](mailto:tasistro@ort.edu.uy)

<sup>4</sup> Email: [szasz@ort.edu.uy](mailto:szasz@ort.edu.uy)

<sup>5</sup> Email: [copello@ort.edu.uy](mailto:copello@ort.edu.uy)

# 1 Introduction

## 1.1 Related Work

There exist several developments in the direction of our work, all of them based the Isabelle/HOL proof assistant. Gordon [2] constructs a similar BVC induction principle over a variation of de Bruijn syntax. The syntax used in Gordons’s work was already suggested by de Bruijn [1], in which “free variables have names but the bound variables are nameless”. In this representation  $\alpha$ -convertible terms are syntactically equal, but invalid terms appears, so a well-formed predicate is needed to eliminate the incorrect terms. Because this last issue, every introduced function must be proved to be closed under well-formed terms. On the other hand, the main advantage of this mixed strategy is that theorems can be expressed in convetional form, without de Bruijn encoding, and in spite of this, the renaming of bound variables is still supported in proofs, because syntactical equality is up to  $\alpha$ -conversion. He hide this explicit renaming from proofs introducing an induction principle for decidable predicates, which is proved by induction on the length of terms. As the BVC convention, this induction principle enable us to choose the abstraction variables fresh enough from the context in a similar way as we will do in this work. Although, as Gordon point outs, we belive name-carrying syntax up to literal equality would be needed to represent language definitions, such as that of standard ML, for instance, where syntax is not identified up to  $\alpha$ -conversion. De Bruijn notation has been used to implement several theorem provers, where syntax is internally represented in De Bruijn but human interacting uses a name-carrying notation. But this is different to use also this internal notation at a logic level.

Previous approach is *first-order* in the sense that the variable-binding operations of the embedded syntax is distinct form variable-binding at the host proof assistant language. In [3], Gordon and Melham began to explore a *second-order* approach ...

## 1.2 A brief introduction to Agda

The Agda system [6] can be seen both as a programming language with dependent types and as an interactive proof assistant. It implements Martin-Löf’s (intentional) type theory [4] and it extends this theory with a number of features that makes programming more convenient. In order to guarantee logical consistency, only functions that can syntactically be checked total can be defined in the system.

The syntax of Agda resembles that of Haskell and provides many standard programming constructor such as modules, datatypes and case-expressions, signatures and records, and let- and where-expressions. It also allows defining functions by pattern matching on one or several of the function’s arguments, and supports a very flexible way of naming expressions and types, including the possibility of having mixfix names (the positions of the arguments in the name are indicated by the character `_`) and of using Unicode in the names.

Agda supports the possibility of abstracting over the result of an expression when defining a function. This abstraction is performed with the constructor `with`, which effectively adds another argument to the function to be defined that can then be matched in the usual way.

It is possible to omit terms that the type checker can figure out for itself by replacing them by `_`. Agda also allows the possibility of defining certain arguments as "implicit", which is done by using brackets in the declarations of the arguments. Implicit arguments do not need to be provided, so only arguments that the system can deduce on its own should be defined as implicit. To give an implicit argument explicit one should embrace the corresponding expression with brackets.

## 2 Infrastructure

```

data  $\Lambda$  : Set where
  v      : Atom  $\rightarrow$   $\Lambda$ 
   $\_ \cdot \_$  :  $\Lambda \rightarrow \Lambda \rightarrow \Lambda$ 
   $\lambda \_$   : Atom  $\rightarrow \Lambda \rightarrow \Lambda$ 

data  $\_ \# \_$  (a : Atom) :  $\Lambda \rightarrow$  Set where
   $\#v$  : {b : Atom}  $\rightarrow b \neq a \rightarrow a \# v b$ 
   $\# \cdot$  : {M N :  $\Lambda$ }  $\rightarrow a \# M \rightarrow a \# N \rightarrow a \# M \cdot N$ 
   $\# \lambda \equiv$  : {M :  $\Lambda$ }  $\rightarrow a \# \lambda a M$ 
   $\# \lambda$  : {b : Atom} {M :  $\Lambda$ }  $\rightarrow a \# M \rightarrow a \# \lambda b M$ 

( $\_ \bullet \_$ )a : Atom  $\rightarrow$  Atom  $\rightarrow$  Atom  $\rightarrow$  Atom
( $a \bullet b$ )a c with c  $\stackrel{?}{=}_a$  a
... | yes _ = b
... | no _ with c  $\stackrel{?}{=}_a$  b
... | yes _ = a
... | no _ = c

( $\_ \bullet \_$ )  : Atom  $\rightarrow$  Atom  $\rightarrow$   $\Lambda \rightarrow \Lambda$ 
( $a \bullet b$ ) v c = v ((a • b)a c)
( $a \bullet b$ ) M • N = ((a • b)a M) • ((a • b)a N)
( $a \bullet b$ )  $\lambda$  c M =  $\lambda$  ((a • b)a c) ((a • b)a M)

 $\_ \bullet_a \_$  :  $\Pi \rightarrow$  Atom  $\rightarrow$  Atom
 $\pi \bullet_a a = \text{foldr } (\lambda s b \rightarrow (\text{proj}_1 s \bullet \text{proj}_2 s)_a b) a \pi$ 

 $\_ \bullet \_$  :  $\Pi \rightarrow \Lambda \rightarrow \Lambda$ 
 $\pi \bullet M = \text{foldr } (\lambda s M \rightarrow (\text{proj}_1 s \bullet \text{proj}_2 s) M) M \pi$ 

data  $\_ \sim_\alpha \_$  :  $\Lambda \rightarrow \Lambda \rightarrow$  Set where
   $\sim_\alpha v$  : {a : Atom}  $\rightarrow v a \sim_\alpha v a$ 
   $\sim_\alpha \cdot$  : {M M' N N' :  $\Lambda$ }  $\rightarrow M \sim_\alpha M' \rightarrow N \sim_\alpha N' \rightarrow M \cdot N \sim_\alpha M' \cdot N'$ 
   $\sim_\alpha \lambda$  : {M N :  $\Lambda$ } {a b : Atom} {xs : List Atom}
     $\rightarrow ((c : \text{Atom}) \rightarrow c \notin xs \rightarrow (a \bullet c) M \sim_\alpha (b \bullet c) N) \rightarrow \lambda a M \sim_\alpha \lambda b N$ 

```

### 3 Induction Principles

Primitive induction over  $\Lambda$  terms.

$$\begin{aligned}
\text{TermPrimInd} : \{l : \text{Level}\} & (P : \Lambda \rightarrow \text{Set } l) \\
& \rightarrow (\forall a \rightarrow P(\mathbf{v} \ a)) \\
& \rightarrow (\forall M \ N \rightarrow P \ M \rightarrow P \ N \rightarrow P \ (M \cdot N)) \\
& \rightarrow (\forall M \ b \rightarrow P \ M \rightarrow P(\mathbf{\lambda} \ b \ M)) \\
& \rightarrow \forall M \rightarrow P \ M
\end{aligned}$$

Fig. 1. Primitive Induction Principle

The next induction principle provides a strong hypothesis for the lambda abstraction case: it namely allows to assume the property for all renamings (given by finite permutations of names) of the body of the abstraction:

$$\begin{aligned}
\text{TermIndPerm} : \{l : \text{Level}\} & (P : \Lambda \rightarrow \text{Set } l) \\
& \rightarrow (\forall a \rightarrow P(\mathbf{v} \ a)) \\
& \rightarrow (\forall M \ N \rightarrow P \ M \rightarrow P \ N \rightarrow P \ (M \cdot N)) \\
& \rightarrow (\forall M \ b \rightarrow (\forall \pi \rightarrow P(\pi \bullet M)) \rightarrow P(\mathbf{\lambda} \ b \ M)) \\
& \rightarrow \forall M \rightarrow P \ M
\end{aligned}$$

Fig. 2. Permutation Induction Principle

Notice that the hypothesis provided for the case of abstractions is akin to the corresponding one of the principle of strong or complete induction on the size of terms, only that expressed in terms of name permutations. This principle can be derived from the former, i.e. from simple structural induction, in very much the same way as complete induction on natural numbers is derived from ordinary mathematical induction, i.e. by using primitive induction to prove that  $\forall \pi, P(\pi \bullet M)$  given also the hypothesis of this new introduced principle. For the interesting abstraction case of the primitive induction proof, we have to prove  $\forall \pi, P(\pi \bullet \mathbf{\lambda} \ a \ M)$ , which is equal to  $\forall \pi, P(\mathbf{\lambda} \ (\pi \bullet_a a) \ (\pi \bullet M))$ . The abstraction hypothesis of the new principle give us that  $\forall M', b, (\forall \pi' \rightarrow P(\pi' \bullet M')) \rightarrow P(\mathbf{\lambda} \ b \ M')$ , instantiating  $M'$  as  $\pi \bullet M$  and  $b$  as  $\pi \bullet_a a$  in previous hypothesis, we obtain the desired result if we know that  $\forall \pi', P(\pi' \bullet \pi \bullet M)$ . Which holds using the primitive inductive hypothesis  $\forall \pi'', P(\pi'' \bullet M)$  with  $\pi'' = \pi' \bullet \pi$ .

A predicate is  $\alpha$ -compatible if it holds for a given term it also holds for all  $\alpha$ -equivalent terms.

$$\begin{aligned}
\alpha\text{CompatiblePred} : \{l : \text{Level}\} & \rightarrow (\Lambda \rightarrow \text{Set } l) \rightarrow \text{Set } l \\
\alpha\text{CompatiblePred } P & = \{M \ N : \Lambda\} \rightarrow M \sim_\alpha N \rightarrow P \ M \rightarrow P \ N
\end{aligned}$$

If a predicate is  $\alpha$ -compatible then we can prove the following induction principle using previously introduced one. This new principle enables us to choose the variable of the abstraction case different from a given finite list of variables. In this way this principle allow us to emulate Barendregt Variable Convention (BVC), and assume enough fresh variables in a proof, that is, doing proofs over  $\alpha$ -equivalence classes of terms. Our aim is use this principle whenever is possible, previous and next principles are usefull to internaly deal with swap operation, but we want to hide this operation from our proofs as much as possible.

$$\text{Term}\alpha\text{PrimInd} : \{l : \text{Level}\} (P : \Lambda \rightarrow \text{Set } l) \rightarrow \alpha\text{CompatiblePred } P$$

$$\begin{aligned}
&\rightarrow (\forall a \rightarrow P(\textcolor{green}{v} a)) \\
&\rightarrow (\forall M N \rightarrow P M \rightarrow P N \rightarrow P(M \cdot N)) \\
&\rightarrow \exists (\lambda vs \rightarrow (\forall M b \rightarrow b \notin vs \rightarrow P M \rightarrow P(\textcolor{green}{\lambda} b M))) \\
&\rightarrow \forall M \rightarrow P M
\end{aligned}$$

Again assuming an  $\alpha$ -compatible predicate, and using the second induction principle (figure 2), we can prove the following induction principle which combines the two previous principles characteristics.

$$\begin{aligned}
\text{Term}\alpha\text{IndPerm} : \{l : \text{Level}\}(P : \Lambda \rightarrow \text{Set } l) &\rightarrow \alpha\text{CompatiblePred } P \\
&\rightarrow (\forall a \rightarrow P(\textcolor{green}{v} a)) \\
&\rightarrow (\forall M N \rightarrow P M \rightarrow P N \rightarrow P(M \cdot N)) \\
&\rightarrow \exists (\lambda as \rightarrow (\forall M b \rightarrow b \notin as \rightarrow (\forall \pi \rightarrow P(\pi \bullet M)) \rightarrow P(\textcolor{green}{\lambda} b M))) \\
&\rightarrow \forall M \rightarrow P M
\end{aligned}$$

Fig. 3. Permutation  $\alpha$ -Induction Principle

## 4 The Choice Function

Our recursion principle will proceed by recursion on the structure of terms. The interesting case is of course that of the  $\lambda$ -abstractions, in which the principle ought to satisfy two requirements:

- (i) To allow defining the function in question by assuming that the abstraction is  $\textcolor{green}{\lambda} \chi N$  where  $\chi$  is a name not belonging to a given finite set of names. (This is Barendregt's convention —in practical cases the set of names to be avoided is determined by a certain predicate but it is always finite.)
- (ii) To equate (i.e. not distinguish)  $\sim_\alpha$ -equivalent terms while at the same time avoiding a too coarse identification. (A principle yielding e.g. only constant functions would indeed identify  $\sim_\alpha$ -equivalent terms, but it would be devoid of any interest.)

It would then seem that we need to know the full definition of  $\sim_\alpha$  in order to implement our recursor. But such is not the case, as what we need in fact is only to determine how to proceed in the case of  $\lambda$ -abstractions, whose behavior with respect to  $\sim_\alpha$  is determined by the simple expedient of “changing the bound atom”. So suppose the procedure is receiving an abstraction  $\textcolor{green}{\lambda} x M$ . In order to implement Barendregt's convention, the finite set of names to be avoided has to be provided to the method, which we may represent as a given list  $vs$  of atoms. Further, we notice that:

$$\chi \# \textcolor{green}{\lambda} x M \Rightarrow \textcolor{green}{\lambda} x M \sim_\alpha \textcolor{green}{\lambda} \chi (x \bullet \chi) M$$

So we obtain an implementation of the required procedure by choosing a name  $\chi$  such that:

- (i)  $\chi \notin vs$
- (ii)  $\chi \# \textcolor{green}{\lambda} x M$

and then stating the value of the function which is being defined which corresponds to input  $\textcolor{green}{\lambda} \chi (x \bullet \chi) M$ . For this, the result of the function on  $(x \bullet \chi) M$  can

be (recursively) used. Therefore we are implementing a form of inductive/recursive reasoning on, so to speak,  $\sim_\alpha$ -equivalence classes: The value of the function that is being defined for any abstraction is determined as the value of a convenient representative of its  $\sim_\alpha$ -equivalence class. Actually, for the defined correspondence to be indeed a function, it is important that the name  $\chi$  is chosen in a deterministic way. This can be easily achieved, for instance by determining  $\chi$  as the *first* name satisfying the requirements above in a fixed enumeration of the type of names, which must of course exist. We implement such determinism coding the following function:

$\chi : \text{List Atom} \rightarrow \Lambda \rightarrow \text{Atom}$

This functions has the desired previously enumerated properties. As a consequence of the imposed fresh property, the application of  $\chi$  function to the same list of variables and  $\sim_\alpha$ -equivalent terms should return the same atom, because  $\sim_\alpha$ -equivalent terms have the same fresh atoms.

At this point our development seems to diverge from the nominal one. Our  $\chi$  function is reduced to an auxiliary one with the following signature:

$\chi' : \text{List V} \rightarrow \text{V}$

which is not *finite supported*, while nominal theory requires functions to be finitely supported. A function  $f : X \rightarrow Y$ , where  $X, Y$  are nominal sets, is finitely supported if there exists a finite set of atoms  $A$  that for all atoms  $a, a' \notin A$  and any  $x \in X$   $(a \bullet a')(f((a \ a')x)) = f(x)$ . Back to our choice  $\chi'$ , it can be seen that there no exists such fixed set of atoms such that for any list of atoms in the image, swapping some atoms in it, then applying our choice function, and then again applying the same swapping have no effect in the result.

## 5 Iteration and Recursion Principles

We want to define strong  $\alpha$ -compatible functions, that is, functions over the  $\alpha$ -equivalence class of terms. So this functions should not depend on the abstraction variables of a term and return the same result between  $\alpha$ -equivalent terms.

$\text{strong}\sim_\alpha\text{Compatible} : \{l : \text{Level}\}\{A : \text{Set } l\} \rightarrow (\Lambda \rightarrow A) \rightarrow \Lambda \rightarrow \text{Set } l$   
 $\text{strong}\sim_\alpha\text{Compatible } f M = \forall N \rightarrow M \sim_\alpha N \rightarrow f M \equiv f N$

We define an iteration principle over raw terms which always produces  $\alpha$ -compatible functions. For the abstraction case, this principle also allow us to give a list of variables from where the abstractions variables will not to be choosen, this will be usefull to define the no capture substitution operation latter. This iteration principle is derived from the BVC induction principle ([Term \$\alpha\$ PrimInd](#)) in a direct manner, where the predicate is a constant function in [Set](#) which always returns the type  $A$ , that is, a trivial constant predicate equivalent to  $A \neq \emptyset$ , so the proof of  $\alpha$ -compatibililty of this also trivial. Note the application and abstraction cases drop the recursive sub-terms, so we could have defined a recursion principle if we have used this terms, but this will give us a not strong compatible recursion principle. This iteration implementation only uses the recursive calls, and in the abstraction case, the new abstraction variable chossen fresh from the given list of variables and

the inspected term.

$$\begin{aligned}
\text{Alt} &: \{l : \text{Level}\}(A : \text{Set } l) \\
&\rightarrow (\text{Atom} \rightarrow A) \\
&\rightarrow (A \rightarrow A \rightarrow A) \\
&\rightarrow \text{List Atom} \times (\text{Atom} \rightarrow A \rightarrow A) \\
&\rightarrow \Lambda \rightarrow A \\
\text{Alt } A \text{ hv } h \cdot (vs , h\lambda) &= \text{Term}\alpha\text{PrimInd } (\lambda \_ \rightarrow A) \\
&\quad (\lambda \_ \rightarrow \text{id}) \\
&\quad hv \\
&\quad (\lambda \_ \_ \rightarrow h \cdot) \\
&\quad (vs , (\lambda \_ b \_ r \rightarrow h\lambda b r))
\end{aligned}$$

Next result make explicit the iterator behaviour in the abstraction case.

$$\begin{aligned}
\text{Alt}\lambda &: \{l : \text{Level}\}(A : \text{Set } l) \\
&\rightarrow (hv : \text{Atom} \rightarrow A) \\
&\rightarrow (h \cdot : A \rightarrow A \rightarrow A) \\
&\rightarrow (vs : \text{List Atom}) \\
&\rightarrow (h\lambda : \text{Atom} \rightarrow A \rightarrow A) \\
&\rightarrow \forall a M \\
&\rightarrow \text{Alt } A \text{ hv } h \cdot (vs , h\lambda) (\lambda a M) \equiv \\
&\quad h\lambda (\lambda vs (\lambda a M)) \\
&\quad (\text{Alt } A \text{ hv } h \cdot (vs , h\lambda) ([a , (\lambda vs (\lambda a M))] \bullet M))
\end{aligned}$$

The following lemma says our iteration principle always return strong compatibility functions. This result is proved using the induction principle in figure 3.

$$\begin{aligned}
\text{lemmaAltStrong}\alpha\text{Compatible} &: \{l : \text{Level}\}(A : \text{Set } l) \\
&\rightarrow (hv : \text{Atom} \rightarrow A) \\
&\rightarrow (h \cdot : A \rightarrow A \rightarrow A) \\
&\rightarrow (vs : \text{List Atom}) \\
&\rightarrow (h\lambda : \text{Atom} \rightarrow A \rightarrow A) \\
&\rightarrow (M : \Lambda) \rightarrow \text{strong}\sim\alpha\text{Compatible } (\text{Alt } A \text{ hv } h \cdot (vs , h\lambda)) M
\end{aligned}$$

Fig. 4. Strong  $\alpha$  Compatibility of the Iteration Principle

From this iteration principle we directly derive the next recursion principle over terms, which also generates strong  $\alpha$ -compatible functions.

$$\begin{aligned}
\Lambda\text{Rec} &: \{l : \text{Level}\}(A : \text{Set } l) \\
&\rightarrow (\text{Atom} \rightarrow A) \\
&\rightarrow (A \rightarrow A \rightarrow \Lambda \rightarrow \Lambda \rightarrow A) \\
&\rightarrow \text{List Atom} \times (\text{Atom} \rightarrow A \rightarrow \Lambda \rightarrow A) \\
&\rightarrow \Lambda \rightarrow A
\end{aligned}$$

## 6 Iterator Application

We present several applications of the iteration/recursive principle defined in previous section. In the following two sub-sections we implement two classic examples of

$\lambda$ -calculus theory. In the appendix A we also apply our iteration/recursion principle to the examples of functions over terms presented in [5]. This work presents a sequence of increasing complexity functions, providing this way a set of functions to test recursion principles over  $\lambda$ -calculus terms. Each of the defined functions respects the  $\alpha$ -equivalence relation, that is, are strong compatible functions by being implemented over the previously introduced iteration/recursion principles.

### 6.1 Free Variables

We implement the function that returns the free variables of a term.

```
fv :  $\Lambda \rightarrow \text{List Atom}$ 
fv = Alt (List Atom) [ ] _++_ ([ ] ,  $\lambda v r \rightarrow r - v$ )
```

As a direct consequence of strong  $\alpha$ -compatibility of the iteration principle we have that  $\alpha$  compatible terms have equal free variables.

The relation  $\_*$  holds when a variables occurs free in a term.

```
data _*_ : Atom  $\rightarrow \Lambda \rightarrow \text{Set}$  where
  *v : {x : Atom}  $\rightarrow x * v$ 
  *.l : {x : Atom} {M N :  $\Lambda$ }  $\rightarrow x * M \rightarrow x * (M \cdot N)$ 
  *.r : {x : Atom} {M N :  $\Lambda$ }  $\rightarrow x * N \rightarrow x * (M \cdot N)$ 
  * $\lambda$  : {x y : Atom} {M :  $\Lambda$ }  $\rightarrow x * M \rightarrow y \not\equiv x \rightarrow x * (\lambda y M)$ 
```

We can use the last induction principle (fig. 3) to prove the following proposition:

```
Pfv* : Atom  $\rightarrow \Lambda \rightarrow \text{Set}$ 
Pfv* a M = a  $\in$  fv M  $\rightarrow a * M$ 
```

In the  $\lambda$ - abstraction case of the induction proof, we can exclude the variable  $a$  from the abstraction variables of the term where the induction is done, simplifying this proof. We have to prove that  $\forall b \not\equiv a, a \in \text{fv}(\lambda b M) \Rightarrow a * \lambda b M$ , knowing by inductive hypothesis that  $\forall \pi, a \in \text{fv}(\pi \bullet M) \Rightarrow a * (\pi \bullet M)$ . Using the lemma  $\text{Alt}\lambda$ , about the behaviour of the iterator for the abstraction case, we know  $\text{fv}(\lambda b M) = \text{fv}((b \bullet \chi) M) - \chi$  where  $\chi = \lambda [] \lambda b M$ , so we can infer that  $a \in \text{fv}((b \bullet \chi) M)$  and  $a \not\equiv \chi$ . We can use the inductive hypothesis with  $\pi = [(b, \chi)]$  and previous result to obtain that  $a * ((b \bullet \chi) \bullet M)$ , which using that  $b \not\equiv a$  and  $a \not\equiv \chi$  we can obtain that  $a * M$ . Finally, we are able to apply the constructor  $\lambda^*$  of the relation  $*$  to previous result and  $b \not\equiv a$  to obtain the desired result.

The flexibility to exclude variable  $a$  comes with an extra cost, we need to prove that the predicate  $\forall a, \text{Pfv}^* a$  is  $\alpha$ -compatible in order to use the choosen induction principle. This  $\alpha$ -compatible proof is direct once we prove that  $*$  is an  $\alpha$ -compatible relation and the  $\text{fv}$  function is strong  $\alpha$ -compatible. The last property is direct because we implemented  $\text{fv}$  with the iteration principle, so the extra cost is just the proof that  $*$  is  $\alpha$ -compatible.

Another approach where the last proof can be automatically obtained, as we freely obtained that  $\text{fv}$  is strong  $\alpha$ -compatible, is to define the free relation using our iteration principle, and not a data type as previously done.

```
_free_ : Atom  $\rightarrow \Lambda \rightarrow \text{Set}$ 
```



$$(\text{free } \_) a = \text{Alt Set } (\lambda b \rightarrow a \equiv b) \_ \vee \_ ([a], \lambda \_ \rightarrow \text{id})$$

For the variable case we return the type of the propositional equality, inhabited only if the searched variable is equal to the term variable. The application case is the disjoint union of the types returned by the recursive calls, that is, the union of the variable free occurrence evidence in the applied terms. Finally, in the abstraction case we can choose the abstraction variable to be different from the searchhead one, so we can ignore the abstraction variable, and return just the recursive call, containing the evidence of any variable free occurrence in the abstraction body.

This free predicate impementation is strong compatible by construction because we build it from our iterator principle, so given any variable  $a$  and two  $\alpha$ -compatible terms  $M, N$ , the returned set should be the same. So is direct that if the predicate holds for  $M$  (which means that the returned set is inhabited for  $M$ ), then the predicate should also hold for  $N$ .

From this point we can do an analog proof of  $\text{Pfv}^*$  proposition, but now using this new free predicate definition which is  $\alpha$ -compatible by construction. This give us a framework where we can define strong compatible functions and also  $\alpha$ -compatible predicates over terms, and then prove properties about theses functions and predicates using our induction principle that provides us with the BVC.

## 6.2 Substitution

We implement the no capture substitution operation, we avoid any variable capture giving as variables to not to choose from as variable abstractions: the substituted variable and the free variables of the replaced term.

$$\begin{aligned} \text{hvar} &: \text{Atom} \rightarrow \Lambda \rightarrow \text{Atom} \rightarrow \Lambda \\ \text{hvar } x \ N \ y \ \text{with } x &\stackrel{?}{=} a \ y \\ \dots \mid \text{yes } \_ &= N \\ \dots \mid \text{no } \_ &= v \ y \\ - \\ \_ [\_ := \_] &: \Lambda \rightarrow \text{Atom} \rightarrow \Lambda \rightarrow \Lambda \\ M [a := N] &= \text{Alt } \Lambda (\text{hvar } a \ N) \_ \_ (a :: \text{fv } N, \lambda) M \end{aligned}$$

Again because of the strong  $\alpha$ -compability of the iteration principle we obtain the following result for free:

$$\begin{aligned} \text{lemmaSubst1} &: \{M \ N : \Lambda\} (P : \Lambda) (a : \text{Atom}) \\ &\rightarrow M \sim_{\alpha} N \rightarrow M [a := P] \equiv N [a := P] \end{aligned}$$

Using the induction principle in figure 2 we prove:

$$\begin{aligned} \text{lemmaSubst2} &: \forall \{N\} \{P\} \ M \ x \\ &\rightarrow N \sim_{\alpha} P \rightarrow M [x := N] \sim_{\alpha} M [x := P] \end{aligned}$$

From the two previous result we directly obtain next  $\alpha$ -compatibility substitution lemma .

$$\text{lemmaSubst} : \{M \ N \ P \ Q : \Lambda\} (a : \text{Atom})$$

```

→ M ~α N → P ~α Q
→ M [ a := P ] ~α N [ a := Q ]
lemmaSubst {M} {N} {P} {Q} a M~N P~Q
= begin
  M [ a := P ]
  ≈⟨ lemmaSubst1 P a M~N ⟩
  N [ a := P ]
  ~⟨ lemmaSubst2 N a P~Q ⟩
  N [ a := Q ]
□

```

With previous result we can derive that our substitution operation is  $\alpha$ -equivalent with a naive one for fresh enough abstraction variables.

```

lemmaλ~[] : ∀ {a b P} M → b ∉ a :: fv P
→ λ b M [ a := P ] ~α λ b (M [ a := P ])

```

We can combine this last result with the `TermαPrimInd` principle which emulates BVC convention, and mimic in this way a pen and pencil inductive proofs over  $\alpha$ -equivalence classes of terms about substitution operation. As an example we show next substitution composition predicate:

```

PSC : ∀ {x y L} N → Λ → Set
PSC {x} {y} {L} N M = x ≠ y → x ∉ fv L
→ (M [ x := N ]) [ y := L ] ~α (M [ y := L ]) [ x := N [ y := L ] ]

```

Next we give a direct equational proof that `PSC` predicate is  $\alpha$ -compatible:

```

αCompatiblePSC : ∀ {x y L} N → αCompatiblePred (PSC {x} {y} {L} N)
αCompatiblePSC {x} {y} {L} N {M} {P} M~P PM x≠y x∉fvL
= begin
  (P [ x := N ]) [ y := L ]
  - Strong α compability of inner substitution operation
  ≈⟨ cong (λ z → z [ y := L ]) (lemmaSubst1 N x (σ M~P)) ⟩
  (M [ x := N ]) [ y := L ]
  - We apply that we know the predicate holds for M
  ~⟨ PM x≠y x∉fvL ⟩
  (M [ y := L ]) [ x := N [ y := L ] ]
  - Strong α compability of inner substitution operation
  ≈⟨ cong (λ z → z [ x := N [ y := L ] ]) (lemmaSubst1 L y (M~P)) ⟩
  (P [ y := L ]) [ x := N [ y := L ] ]
□

```

For the interesting abstraction case of the  $\alpha$ -structural induction over the lambda term, we assume the abstraction variables in the term are not in the substituted variables nor the substituted terms. In this way the substitution operations are  $\alpha$ -compatible to naive substitutions, then the inductive hypothesis allow us to complete the the inductive proof in a direct maner. Next we show the code fragment correspondint to this proof:

```

begin
  (λ b M [ x := N ] ) [ y := L ]
  - Inner substitution is α equivalent
  - to a naive one because b ∉ x :: fv N
  ≈⟨ lemmaSubst1 L y (lemmaλ~[] M b∉x::fvN) ⟩
    (λ b (M [ x := N ])) [ y := L ]
  - Outer substitution is α equivalent
  - to a naive one because b ∉ y :: fv L
  ~⟨ lemmaλ~[] (M [ x := N ]) b∉y::fvL ⟩
    λ b ((M [ x := N ]) [ y := L ])
  - We can now apply our inductive hypothesis
  ~⟨ lemma~αλ (IndHip x≠y x∉fvL) ⟩
    λ b ((M [ y := L ]) [ x := N [ y := L ] ])
  - Outer substitution is α equivalent
  - to a naive one because b ∉ x :: fv N [y := L]
  ~⟨ σ (lemmaλ~[] (M [ y := L ]) b∉x::fvN[y:=L]) ⟩
    (λ b (M [ y := L ])) [ x := N [ y := L ] ]
  - Inner substitution is α equivalent
  - to a naive one because b ∉ y :: fv L
  ≈⟨ sym (lemmaSubst1 (N [ y := L ]) x (lemmaλ~[] M b∉y::fvL)) ⟩
    (λ b M [ y := L ]) [ x := N [ y := L ] ]
  □

```

Remarkably theses results are directly derived from the first primitive induction principle, and no need of induction on the length of terms or accesible predicates were needed in all of this formalization.

## A Iteration/Recursion Applications

In the following sections we successfully apply our iteration/recursion principle to all the examples from [5]. This work presents a sequence of increasing complexity functions definitions to provide a test for any principle of function definition, where each of the given functions respects the  $\alpha$ -equivalence relation.

### A.1 Case Analysis and Examining Constructor Arguments

The following family of functions distinguishes between constructors returning the constructor components, giving in a sense a kind of *pattern-matching*.

$$\begin{array}{ll}
 isVar : \Lambda \rightarrow \text{Maybe (Variable)} & isApp : \Lambda \rightarrow \text{Maybe } (\Lambda \times \Lambda) \\
 isVar (v \ x) = Just & isApp (v \ x) = Nothing \\
 isVar (M \cdot N) = Nothing & isApp (M \cdot N) = Just(M, N) \\
 isVar (\lambda x M) = Nothing & isApp (\lambda x M) = Nothing
 \end{array}$$

$$isAbs : \Lambda \rightarrow \text{Maybe } (Variable \times \Lambda)$$

$$isAbs (v \ x) = \text{Nothing}$$

$$isAbs (M \cdot N) = \text{Nothing}$$

$$isAbs (\lambda x M) = \text{Just}(x, M)$$

Next we present the corresponding codifications using our iteration/recursion principle:

```

isVar :  $\Lambda \rightarrow \text{Maybe Atom}$ 
isVar =  $\Lambda\text{lt}$  ( $\text{Maybe Atom}$ )
      just
      ( $\lambda \_ \_ \rightarrow \text{nothing}$ )
      ( $[\ ] , \lambda \_ \_ \rightarrow \text{nothing}$ )
-
isApp :  $\Lambda \rightarrow \text{Maybe } (\Lambda \times \Lambda)$ 
isApp =  $\Lambda\text{Rec}$  ( $\text{Maybe } (\Lambda \times \Lambda)$ )
      ( $\lambda \_ \rightarrow \text{nothing}$ )
      ( $\lambda \_ \_ M \ N \rightarrow \text{just } (M , N)$ )
      ( $[\ ] , \lambda \_ \_ \_ \rightarrow \text{nothing}$ )
-
isAbs :  $\Lambda \rightarrow \text{Maybe } (\text{Atom} \times \Lambda)$ 
isAbs =  $\Lambda\text{Rec}$  ( $\text{Maybe } (\text{Atom} \times \Lambda)$ )
      ( $\lambda \_ \rightarrow \text{nothing}$ ) ( $\lambda \_ \_ \_ \_ \rightarrow \text{nothing}$ )
      ( $[\ ] , \lambda \ a \_ \ M \rightarrow \text{just } (a , M)$ )

```

### A.2 Simple recursion

The size function returns a numeric measurement of the size of a term.

$$size : \Lambda \rightarrow \mathbb{N}$$

$$size (v \ x) = 1$$

$$size (M \cdot N) = size(M) + size(N) + 1$$

$$size (\lambda x M) = size(M) + 1$$

```

size :  $\Lambda \rightarrow \mathbb{N}$ 
size =  $\Lambda\text{lt } \mathbb{N}$  ( $\text{const } 1$ ) ( $\lambda \ n \ m \rightarrow \text{succ } n + m$ ) ( $[\ ] , \lambda \_ \ n \rightarrow \text{succ } n$ )

```

### A.3 Alfa Equality

Next functions decides the  $\alpha$ -equality relation between two terms.

```

equal :  $\Lambda \rightarrow \Lambda \rightarrow \text{Bool}$ 
equal =  $\Lambda\text{lt } (\Lambda \rightarrow \text{Bool})$  vareq appeq ( $[\ ] , \text{abseq}$ )
      where

```

```

vareq : Atom →  $\Lambda$  → Bool
vareq a M with isVar M
... | nothing = false
... | just b =  $\lfloor a \stackrel{?}{=} b \rfloor$ 
appeq : ( $\Lambda$  → Bool) → ( $\Lambda$  → Bool) →  $\Lambda$  → Bool
appeq fM fN P with isApp P
... | nothing = false
... | just (M', N') = fM M'  $\wedge$  fN N'
abseq : Atom → ( $\Lambda$  → Bool) →  $\Lambda$  → Bool
abseq a fM N with isAbs N
... | nothing = false
... | just (b, P) =  $\lfloor a \stackrel{?}{=} b \rfloor \wedge fM P$ 

```

Observe that `isAbs` function also normalises  $N$ , so it is correct in last line to ask if the two binders are equal.

#### A.4 Recursion Mentioning a Bound Variable

The *enf* function is true of a term if it is in  $\eta$ -normal form, the *fv* function returns the set of a term's free variables and was previously defined.

```

enf :  $\Lambda$  → Bool
enf (v x) = True
enf (M · N) = enf(M)  $\wedge$  enf(N) + 1
enf ( $\lambda x M$ ) = enf(M)  $\wedge$  ( $\exists N, x/isApp(M) == Just(N, v x) \Rightarrow x \in fv(N)$ )

_⇒_ : Bool → Bool → Bool
false ⇒ b = true
true ⇒ b = b
-
enf :  $\Lambda$  → Bool
enf =  $\Lambda$ Rec Bool (const true) ( $\lambda b1 b2 \_ \_ \rightarrow b1 \wedge b2$ ) ([], absenf)
where
absenf : Atom → Bool →  $\Lambda$  → Bool
absenf a b M with isApp M
... | nothing = b
... | just (P, Q) = b  $\wedge$  (equal Q (v a)  $\Rightarrow a \in b$  (fv P))

```

#### A.5 Recursion with an Additional Parameter

Given the ternary type of possible directions to follow when passing through a term (*Lt*, *Rt*, *In*), corresponding to the two sub-terms of an application constructor and the body of an abstraction, return the set of paths (lists of directions) to the occurrences of the given free variable in a term, where *cons* insert an element in

front of a list.

$$\begin{aligned}
vposns &: Variable \times \Lambda \rightarrow List (List Direction) \\
vposns (x, v y) &= if (x == y) then [[]] else [] \\
vposns (x, M \cdot N) &= map (cons Lt) (vposns x M) ++ \\
&\quad map (cons Rt) (vposns x N) \\
x \neq y \Rightarrow vposns (x, \lambda y M) &= map (cons In) (vposns x M)
\end{aligned}$$

Notice how the condition guard of the abstraction case is translated to the list of variables from where not to chose from the abstraction variable.

```

data Direction : Set where
  Lt Rt In : Direction
-
vposns : Atom → Λ → List (List Direction)
vposns a = Alt (List (List Direction)) varvposns appvposns ([ a ] , absvposns)
  where
    varvposns : Atom → List (List Direction)
    varvposns b with a  $\stackrel{?}{=}_a$  b
    ... | yes _ = [ [] ]
    ... | no _ = []
    appvposns : List (List Direction) → List (List Direction)
    appvposns l r = map (_ :: _ Lt) l ++ map (_ :: _ Rt) r
    absvposns : Atom → List (List Direction) → List (List Direction)
    absvposns a r = map (_ :: _ In) r

```

#### A.6 Recursion with Varying Parameters and Terms as Range

A variant of the substitution function, which substitutes a term for a variable, but further adjusts the term being substituted by wrapping it in one application of the variable named "0" per traversed binder.

$$\begin{aligned}
sub' &: \Lambda \times Variable \times \Lambda \rightarrow \Lambda \\
sub' (P, x, v y) &= if (x == y) then P else (v y) \\
sub' (P, x, M \cdot N) &= (sub'(P, x, M)) \cdot (sub'(P, x, N)) \\
\left. \begin{array}{l} y \neq x \wedge \\ y \neq 0 \wedge \\ y \notin fv(P) \end{array} \right\} \Rightarrow sub' (P, x, \lambda y M) &= \lambda y (sub'((v 0) \cdot M, x, M))
\end{aligned}$$

To implement this function with our iterator principle we must change the parameters order, so our iterator principle now returns a function that is waiting the term to be substituted. In this way we manage to vary the parameter through the iteration.

```

hvar : Atom → Atom →  $\Lambda$  →  $\Lambda$ 
hvar x y with  $x \stackrel{?}{=}_a y$ 
... | yes _ = id
... | no _ =  $\lambda \_ \rightarrow (\mathbf{v} \ y)$ 
-
sub' : Atom →  $\Lambda$  →  $\Lambda$  →  $\Lambda$ 
sub' x M P =  $\Lambda$ lt ( $\Lambda$  →  $\Lambda$ )
               (hvar x)
               ( $\lambda \ f \ g \ N \rightarrow f \ N \cdot g \ N$ )
               ( $x :: 0 :: \mathbf{fv} \ P, \lambda \ a \ f \ N \rightarrow \mathbf{\lambda} \ a \ (f \ ((\mathbf{v} \ 0) \cdot N))$ )
               M P

```

## References

- [1] N.G de Bruijn. Lambda calculus notation with nameless dummies, a tool for automatic formula manipulation, with application to the church-rosser theorem. *Indagationes Mathematicae (Proceedings)*, 75(5):381 – 392, 1972.
- [2] Andrew D. Gordon. A Mechanisation of Name Carrying Syntax up to Alpha Conversion. In *Proceedings of Higher Order Logic Theorem Proving and its Applications*, Lecture Notes in Computer Science, pages 414–426, 1993.
- [3] Andrew D. Gordon and Thomas F. Melham. Five axioms of alpha-conversion. In Joakim von Wright, Jim Grundy, and John Harrison, editors, *Theorem Proving in Higher Order Logics, 9th International Conference, TPHOLs'96, Turku, Finland, August 26-30, 1996, Proceedings*, volume 1125 of *Lecture Notes in Computer Science*, pages 173–190. Springer, 1996.
- [4] Bengt Nordstrom, Kent Petersson, and Jan M. Smith. *Programming in Martin-Löf's Type Theory: An Introduction*. Oxford University Press, USA, 0 edition, July 1990.
- [5] Michael Norrish. Recursive function definition for types with binders. In *In Seventeenth International Conference on Theorem Proving in Higher Order Logics*, pages 241–256, 2004.
- [6] The Agda Team. The Agda Wiki. Available at <http://appserv.cs.chalmers.se/users/ulfn/wiki/agda.php>.